

M1 Internship Report : Effect Typing for Links

2 May 2022 – 22 July 2022, The University of Edinburgh

ROBIN JOURDE, École Normale Supérieure de Lyon, France

SUPERVISED BY SAM LINDLEY and DANIEL HILLERSTRÖM, The University of Edinburgh,
United Kingdom

This paper is the report of my internship at the School of Informatics of the University of Edinburgh with Sam Lindley and Daniel Hillerström from the Laboratory for Foundations of Computer Science. My intership took place from 2 May 2022 to 22 July 2022 in the Informatics Forum.



THE UNIVERSITY *of* EDINBURGH
informatics



Laboratory for Foundations
of Computer Science

Authors' addresses: Robin Jourde, École Normale Supérieure de Lyon, France, Robin.Jourde@ens-lyon.fr; Supervised by Sam Lindley, Sam.Lindley@ed.ac.uk; Daniel Hillerström, Daniel.Hillerstrom@ed.ac.uk, The University of Edinburgh, United Kingdom.

CONTENTS

| | |
|------------------------------|----|
| Abstract | 1 |
| Contents | 2 |
| 1 Introduction | 3 |
| 2 Links | 3 |
| 3 Effect handlers | 4 |
| 3.1 Syntax and usage | 4 |
| 3.2 Examples | 5 |
| 3.3 Semantics | 8 |
| 4 Typing | 10 |
| 4.1 Rows | 10 |
| 4.2 Links type system | 10 |
| 4.3 Typing handlers | 11 |
| 4.4 Polymorphism | 12 |
| 5 Contributions | 13 |
| 5.1 Effect aliases | 13 |
| 5.2 Fresh labels | 13 |
| 5.3 Polymorphic operations | 18 |
| 6 Conclusion | 19 |
| References | 20 |
| A Artefact | 22 |
| B Links technicalities | 22 |
| B.1 Comments | 22 |
| B.2 Tuples | 22 |
| B.3 Operation type | 22 |
| B.4 Empty type | 22 |
| B.5 Option type | 23 |
| B.6 Function types | 23 |
| B.7 Computation and handlers | 23 |
| C Typing rules | 23 |
| D Erasing procedure | 24 |

1 INTRODUCTION

The goal of my internship was to improve the effect typing system of the Links language, allowing aliases, avoiding effect pollution and allowing polymorphic operations.

Links (section 2) is an experimental tierless functional language for the web, mainly developed in the University of Edinburgh. Among its numerous features, Links supports effect handlers. Effect handlers (section 3) are a language feature that is becoming more and more widespread. They allow the user to define and compose many functionalities from exceptions to concurrency and probabilistic programming. My work (section 5) focused on the typing system (section 4), especially dedicated to these handlers.

2 LINKS

“ *Links is a functional programming language designed to make web programming easier.* ”

Links website [Webc]

Links [CLWY07] is a tierless functional programming language for the web developed at the University of Edinburgh. Typical web development involves the manipulation of several *tiers* which means several, sometimes really different, programming languages :

- ◊ the server side written in Python or Java for instance
- ◊ the client side in JavaScript and HTML/CSS
- ◊ the database queries in SQL

This is neither elegant nor really practical and prevents easy binding between different parts of the program that need to communicate data. It can cause *impedance mismatch*, ie situations where the objects required by the server do not match what the database provide and vice versa.

Links avoids these by providing the programmer one single language for each tier (HTML/CSS are still needed) that is compiled into bytecode for the server, JavaScript for the client and SQL for the queries.

Links also supports numerous useful features including

- ◊ polymorphic variants
- ◊ regular expressions
- ◊ session types
- ◊ row typing
- ◊ effect handlers (see 3)

It is used as an experimental language by researchers to try out new ideas and what could be future widespread programming languages features.

Links is strict, statically typed and use a Hindley-Milner type system as well as a Rémy style row typing system 4.1 [Ré94]. Its syntax is inspired from JavaScript and ML. Its compiler [Webb] is written in OCaml.

Remark 2.1. In appendix B, the reader could find some technical points on Links syntax and semantics that are not worth it to develop here. He is invited to look at it whenever needed.

3 EFFECT HANDLERS

In functional programming, functions are seen as black boxes that takes an input and computes an output. Though this view of *pure* functions is convenient and pleases λ -calculus fans, it fails to accurately represent what real programming is. Computations interact with the outer world by many means :

- ◊ input/output
- ◊ concurrency
- ◊ exceptions
- ◊ choice and non determinism
- ◊ ...

The “outer world” here stand for anything that is not the pure computation, from memory, other threads and user interaction to the evaluation context itself. These interactions not taken into account by the black-box model are called *effects*.

Gordon Plotkin and Matija Pretnar [PP09] introduced a construct that allow *composable* and *customisable* user-defined interpretation of effects : *effect handlers*. They give the programmer direct access to the context, in the form of a first class continuation, often called *resumption*.

Effect handlers can be seen as a generalisation of exception handlers. One can perform an effect in the same way as one could raise an exception and this will transfer the flow control to a handler. It is given the delimited continuation between the point where the effect is performed and where it is handled and can use it

0 times and act as an exception handler

once (linear handler)

twice or more (non-linear handler) to simulate a choice for instance

Effect handlers are being shown a growing industrial interest nowadays. Below are listed some of the biggest projects that uses and implement some kind of effect handlers

- ◊ GitHub code analysis library Semantic (used in more than 25 million repositories)
- ◊ Meta JavaScript UI library React (used in more than 2 million websites)
- ◊ Uber Pyro language (Poutine library) [Webd, Webe]
- ◊ WebAssembly [Weba]

3.1 Syntax and usage

We give the Links syntax alongside with usage of effect handlers and operations. All examples in this report are written in Links.

Inside any segment of code, one can invoke an operation Op with payload p writing `do Op(p)`. This has to be scoped into a handler of this operation. It will take the payload and the resumption and decide what to do.

```

1  handle ( code ) {
2      case <Op(p) => r> -> ...
3      case x -> ...
4  }
```

Listing 1. Handler

Listing 1 give the syntax for a handler for Op , `code` being the code that can perform the operation. When the operation is invoked, the right hand side of line 2 is executed with p the payload and r the resumption bound in context. When `code` returns a value, line 3 is executed with x bound to that value.

If the handler gives the resumption a value v , it acts as if `do Op(p)` returned the value v . If the payload has type t and v has type t' , we say that Op has type $t \Rightarrow t'$.

The type system keeps track of what effects can be performed. A function from type a to b which can performs operations $Op1: t1 \Rightarrow t1', \dots, Opk: tk \Rightarrow tk'$ will have type $(a) \{Op1:t1 \Rightarrow t1', \dots, Opk:tk \Rightarrow tk'\} \rightarrow b$.

3.2 Examples

3.2.1 A first exception handler.

```

1  sig assert_positive : (Int) {Fail:() => Zero|_}-> Int
2  fun assert_positive(x) {
3      if (x>0) x else absurd(do Fail)
4  }
```

Listing 2. Fail invocation

This first example shows a function `assert_positive` that takes an integer and returns it only if it positive, else, it raises the effect `Fail` (for details on `absurd` and the `Zero` type, see B.4). This is the simplest example when one operation is invoked with no payload.

```

1  sig maybe : (() {Fail:() => Zero|e}~> a) -> () {Fail{_|e}~> Maybe(a)
2  fun maybe(m) () {
3      handle(m()) {
4          case <Fail> -> Nothing
5          case x -> Just(x)
6      }
7  }
```

Listing 3. maybe handler

This operation can be handled by handler 3. This is actually an exception handler since the resumption is not used in the right hand side (and thus can be omitted in the left hand side). The *effect signature* is `{Fail: ()=> Zero}`, that means there is only one operation called `Fail` that will be handled.

When the operation `Fail` is invoked, the handler directly returns a value, here `Nothing`. When the computation returns a value, the handler will return that value, embedded in the `Maybe` type (on `Maybe` type, see B.5).

The function `maybe` will take a computation that can perform the effect `{Fail: ()=> Zero|e}` ie the operation `Fail` and possibly other effects, captured by the row variable `e`, and returns something of type `a`. It will execute this computation, performing effects from `e` that should be handled by another handler at some point. In the end, it returns a value of type `Maybe(a)`.

The `Fail{...}` in the effect row indicates a polymorphic presence for the operation. Roughly it means that the operation can or cannot be present after. This is needed since the effect type system enforces that all operations in a row are distinct. `Fail` cannot belong to `e` because of the effect row of the input computation.

```

1  sig h2g2 : () {Fail:() => Zero|_}-> Int
2  fun h2g2 () {
3      assert_positive( -6 * -7 )
4  }
5  sig h2g2' : () {Fail:() => Zero|_}-> Int
6  fun h2g2' () {
7      assert_positive( -6 ) * -7
8  }

```

We can then use the `maybe` function to handle the failures in functions `h2g2` and `h2g2'` :

```

1  links> maybe ( h2g2 ) () ;
2  Just(42) : Maybe (Int)
3  links> maybe ( h2g2' ) () ;
4  Nothing : Maybe (Int)

```

Listing 4. Links console

3.2.2 Drunk coin tossing. We consider now a new operation `Choose` that can produce a boolean and use it to simulate a drunk coin toss. The drunk tosses the coin and either let it fall or catch it and see heads or tail.

```

1  typename Toss = [| Heads | Tail |] ;
2
3  sig toss : () {Choose:Bool|_}-> Toss
4  fun toss () {
5      if (do Choose) Heads else Tail
6  }
7

```

```

8  sig drunkToss : () {Choose:Bool, Fail:Zero|_}-> Toss
9  fun drunkToss () {
10     if (do Choose)
11         # catch
12         toss ()
13     else
14         # fall
15         absurd(do Fail)
16 }
17
18 sig drunkTosses : (Int) -> Comp([Toss], {Choose:Bool, Fail:Zero|_})
19 fun drunkTosses (n) () {
20     if (n<=0)
21         []
22     else
23         drunkToss() :: drunkTosses (n-1) ()
24 }

```

Listing 5. Drunk toss

We can define several handlers for this operation.

```

1  # linear handler : always true
2  sig h_true : (Comp(a,{Choose:Bool|e})) -> Comp(a,{Choose{_|e}})
3  fun h_true (m) () {
4      handle (m()) {
5          case <Choose => r> -> r(true)
6      }
7  }
8
9  # non-linear handler : all choices in a list
10 sig h_all : (Comp(a,{Choose:Bool|e})) -> Comp([a],{Choose{_|e}})
11 fun h_all (m) () {
12     handle (m()) {
13         case <Choose => r> -> r(true) ++ r(false)
14         case x -> [x]
15     }
16 }

```

Listing 6. Toss handlers

And then we can compose the different handlers in several ways.

```

1  links> maybe ( h_true ( drunkTosses(2) ) ) () ;

```

```

2 Just([Heads, Heads]) : Maybe ([Toss])
3 links> h_true ( maybe ( drunkTosses(2) ) ) () ;
4 Just([Heads, Heads]) : Maybe ([Toss])
5
6 links> maybe ( h_all ( drunkTosses(2) ) ) () ;
7 Nothing : Maybe ([[Toss]])
8 links> h_all ( maybe ( drunkTosses(2) ) ) () ;
9 [Just([Heads, Heads]), Just([Heads, Tail]), Nothing, Just([Tail, Heads]), Just([Tail,
   Tail]), Nothing, Nothing] : [Maybe ([Toss])]

```

We try different combinations for 2 coin tosses. In the two first cases, there is no failure so the order of handlers won't matter. The coin is always catch and always on heads.

In the third case, `h_all` imposes that all possibilities are explored so at some point, the coin will fall. This causes a failure, catch by `maybe` that simply returns `Nothing`.

The last case is the most complicated one. The failures are catch by `maybe` *under* `h_all`. That means `h_all` won't see failures and will take each result, whether it is a `Nothing` or a list of heads/tail and form its big list of all possibilities.

3.3 Semantics

We give the operational semantics for handlers.

$$\begin{aligned}
 \text{handle } (v) \{ H \} &\longrightarrow N[v/x] \\
 \text{handle } (\mathcal{E}[\text{do } Op_i(v)]) \{ H \} &\longrightarrow N_i[v/p, \text{fun } (x) \{ \text{handle } (\mathcal{E}[x]) \{ H \} \}/r] \\
 &\quad Op_i \notin \mathcal{E}
 \end{aligned}$$

$$\text{where } H = \begin{cases} \text{case } x \mapsto N \\ \text{case } \langle Op_i(p) \Rightarrow r \rangle \mapsto N_i \\ \dots \end{cases}$$

and \mathcal{E} is an evaluation context : $\mathcal{E} ::= [] \mid \text{handle } (\mathcal{E}) \{ H \} \mid \text{var } x = \mathcal{E} ; N$.

$Op \in \mathcal{E}$ means there is a handler for Op in \mathcal{E} .

If there is a value v inside the handler, we call the return case N with x bound to v . When the handler is the most inner one that can handle the operation Op_i invoke with some payload v , what is insured by $Op_i \notin \mathcal{E}$, we bind the payload and build the resumption before calling the case N_i .

The resumption is embedded in the current handler since operation Op_i could be invoked in the rest of the evaluation context \mathcal{E} . This is called a *deep* handler.

Shallow handlers. Links also support *shallow* handlers, ie handlers that do not automatically reuse the handler in the resumption. The programmer has to write it and can choose another handler if needed. Below

is the new rule for shallow handler.

$$\text{shallowhandle } (\mathcal{E}[\text{do } Op_i(v)]) \{ H \} \longrightarrow N_i[v/p, \text{fun } (x) \{ E[x] \}/r] \quad Op_i \notin \mathcal{E}$$

Parameterised handlers. Another useful feature of handlers is that they could be parameterised. The programmer can give the handler a parameter that is carried along the computation.

```

1  sig h_state : (a) -> (Comp(b,{ Get:()=>a, Put:(a)=>() | e }) ~> Comp((b,a), { Get{_,
    Put{_| e }}
2  fun h_state (init) (m) () {
3    handle (m()) (s <- init) {
4      case x -> (x, s)
5      case <Get => r> -> r (s, s)
6      case <Put(s') => r> -> r ((), s')
7    }
8  }
```

Listing 7. Example of parameterised handler

```

1  fun ex () {
2    var x = do Get * 10 ;
3    var x = 2*x + do Get ;
4    do Put(x) ;
5    ()
6  }
7
8  links> h_state (2) (ex) () ;
9  ((), 42) : ((), Int)
```

A second parameter is given to the `handle` construct that binds a variable and the resumption will take an additional parameter.

In the example above, we use that feature to write a state-cell handler. There are two operations `Put` and `Get` that respectively write into the cell and read what is inside. The parameter here is just the content of the cell.

Here are the reduction rules for parameterised handlers

$$\begin{aligned} \text{handle } (v) (s \leftarrow w) \{ H \} &\longrightarrow N[v/x, w/s] \\ \text{handle } (\mathcal{E}[\text{do } Op_i(v)]) &\longrightarrow N_i[v/p, w/s, \\ (s \leftarrow w) \{ H \} &\quad \text{fun } (x, s') \{ \text{handle } (\mathcal{E}[x]) (s \leftarrow s') \{ H \} \}/r] \\ &\quad Op_i \notin \mathcal{E} \end{aligned}$$

Remark 3.1. A parameterised has to be deep since the parameter needs a handler to carry it.

Relationships between handlers. Those three kinds of handlers are equivalent in the sense that it is possible to express each of them using one other.

If it is easy to see how to express a deep handler from a parameterised handler or from a shallow handler, the converse is not so obvious. See [HLA20], section 4 or [Hil22], section 6.

4 TYPING

Links is a strongly statically typed language. In this section, we describe some main features of Links type system that allow us to type effect handlers.

4.1 Rows

In order to type effects, we store in function types the effects that are *allowed* to be perform during the execution of that function. To do so we use rows. They are an unordered and labeled data and type structure. They are also used in others features of Links as records and variants.

In the most general setting, a row is defined by

$$R ::= \bullet \mid \rho \mid l : \delta; R$$

where ρ is a row variable, l is a label and δ is a data. \bullet denotes the empty row.

A row is said to be *open* if it ends with a row variable and *closed* if it ends with the empty row.

For actual records, the data are actual values of the language but in the typing setting (typing effects, records and variants), δ will be a presence (see 4.2).

We require also that in one row, labels are all distinct and we consider row equal up to reordering of pairs labels/data.

Example 4.1. $Fail : Zero; Choose : Bool; \bullet$ is a valid row and is equal to $Choose : Bool; Fail : Zero; \bullet$ but $Fail : Zero; Fail : Int; e$ and $Fail : Zero; Fail : Zero; e$ are not valid. In the open row $Choose : Bool; e$, e could not be substituted by anything with label *Choose*.

4.2 Links type system

We describe in this section and the following Links type system. In order to avoid useless complexity, we focus on the parts of the type system that are useful to understand effect typing. We omit easy and common parts as well as some parts used only for other Links features.

Below is the type and kind syntax.

| | | | | |
|-----------------|-----------|-------|--|---------------------|
| Kinds | K | $::=$ | $Type \mid Row \mid Presence$ | |
| | | | $\mid \kappa$ | kind variable |
| Rows | R, E | $::=$ | \bullet | |
| | | | $\mid l : P; R$ | |
| | | | $\mid \rho$ | row variable |
| Presence | P | $::=$ | $-$ | absent |
| | | | $\mid T$ | present with a type |
| | | | $\mid \theta$ | presence variable |
| Types | T, A, B | $::=$ | P | primitive type |
| | | | $\mid T \rightarrow_E T$ | function type |
| | | | $\mid T \Rightarrow T$ | operation type |
| | | | $\mid (R)$ | record type |
| | | | $\mid [R]$ | variant type |
| | | | $\mid \forall \alpha :: K. T$ | polymorphic type |
| | | | $\mid A B$ | type application |
| | | | $\mid \alpha$ | type variable |
| Primitive types | P | $::=$ | $Int \mid Bool \mid String \mid \dots$ | |

We have three different kinds that are usual *types*, *rows* and *presence types*.

Any label that appears into a row is associated with a presence type that indicates if the label is present in this row. If it is, we have to provide a type, otherwise this is not necessary and we just indicate the absence. One may wonder why are presence useful. After all, to know if a label is present into a row, looking at the row should be enough. This is true for closed rows but not for open ones. Writing $l : -; \rho$ implies that the row variable ρ cannot stand for a row containing l since a label cannot appear twice into a row. Absence guaranties that whatever the row variable is substituted with, this label will never occur.

Remark 4.2. In *closed* rows, absence is redundant :

$$l_0 : -; l_1 : P_1; \dots; l_k : P_k; \bullet = l_1 : P_1; \dots; l_k : P_k; \bullet$$

Links syntax for presence and effect rows. In Links, we write effect rows between braces $\{ \dots \}$. For a label A , the absence is denoted $A-$, the typed presence $A : \text{type}$ and the polymorphic presence $A\{p\}$, with p a variable of presence kind. The empty row is denoted by $\{\}$, the closed row $l_1 : \delta_1; \dots; l_k : \delta_k$ is denoted by $\{l_1:d_1, \dots, l_k:d_k\}$ and the open row $l_1 : \delta_1; \dots; l_k : \delta_k; e$ is denoted by $\{l_1:d_1, \dots, l_k:d_k \mid e\}$.

4.3 Typing handlers

The typing judgment has to take into account the current allowed effects. We gather them into a row R and we write $\Delta; \Gamma \vdash e : T/R$ for expression e has type T in context $\Delta; \Gamma$ and can perform effects in R .

Remark 4.3. The context contains two parts Γ and Δ . $\Gamma = x : A; \dots$ contains typed term variables and $\Delta = \alpha :: K; \dots$ kinded type variables.

We give below the main typing rules that allow to understand how effect handlers should be typed. The other rules are the obvious ones and are not mentioned. For the sake of clarity, shallow and parameterised handlers rules have also been moved to the appendix C.

$$\begin{array}{c}
\frac{\Delta; \Gamma \vdash p : B/R \quad R = l : B \Rightarrow A; R'}{\Delta; \Gamma \vdash \text{do } l(p) : A/R} \text{Op} \quad \frac{\Delta; \Gamma \vdash f : B \rightarrow_R A/R \quad \Delta; \Gamma \vdash t : B/R}{\Delta; \Gamma \vdash ft : A/R} \text{App} \\
\\
\frac{\Delta; \Gamma, x : B \vdash e : A/R}{\Delta; \Gamma \vdash \text{fun}(x)\{e\} : B \rightarrow_R A/R} \text{Fun} \quad \frac{\Delta; \Gamma \vdash e : A/R \quad R \subseteq R'}{\Delta; \Gamma \vdash e : A/R'} \text{WeakEff} \\
\\
\frac{\Delta; \Gamma \vdash m : B/R' \quad \Delta; \Gamma, x : B \vdash n : A/R \quad \begin{array}{c} R' = l_i : B_i \Rightarrow A_i; R_0 \quad R = l_i : \theta; R_0 \\ \Delta; \Gamma, p_i : B_i, r_i : A_i \rightarrow_R A \vdash n_i : A/R \end{array}}{\Delta; \Gamma \vdash \text{handle}(m)\{\text{case } x \mapsto n \text{ case } \langle l_i(p_i) \Rightarrow r_i \rangle \mapsto n_i\} : A/R} \text{Handle}
\end{array}$$

where $R \subseteq R'$ is defined by

$$\frac{}{R \subseteq R} \quad \frac{}{R \subseteq \rho} \quad \frac{R \subseteq R'}{R \subseteq l : P; R'} \quad \frac{R \subseteq R'}{l : P; R \subseteq l : P; R'} \quad \frac{R \subseteq R'}{l : \theta; R \subseteq l : P; R'}$$

Rule *Op* states that invoking an operation requires an ambient effect row that contains it with the correct types. Rule *App* ensures that, when applying a function, its effect row is compatible with the ambient one. The *Handle* rule types a `handle` construct. The important thing to notice is that the row used to type m obviously contains the operations l_i handled and that after handling, all l_i s have been removed from the row. Finally *WeakEff* ensures that the ambient effect row mentions effects that are *allowed* to be performed but not necessary since an effect can be added to the row.

We say that an expression e type-checks only if no unhandled effects are allowed : $\Delta; \Gamma \vdash e : T/\bullet$, noted $\Delta; \Gamma \vdash e : T$.

Theorem 4.4. *If $\Delta; \Gamma \vdash e : T/R$, in the execution of e only effects in R are performed and not handled.*

Corollary 4.5. *If $\Delta; \Gamma \vdash e : T$, in the execution of e all effects are handled.*

See [HL16] for the proof.

4.4 Polymorphism

Links supports a System F [GTL89] style polymorphism. Types can contain kinded variables α that can be abstracted over with a $\forall \alpha :: K$. A polymorphic type can be instantiated with a type application.

For more details on Links polymorphism, one could look at [CCFL19].

5 CONTRIBUTIONS

During my internship I added some new features to Links that have been merged into the 0.0.0 release.

5.1 Effect aliases

In Links we can define type aliases using the keyword `typename`.

```

1  # sum type via variants
2  typename BinTree(a) = [| Node(BinTree(a), a, BinTree(a)) | Leaf |] ;
3  # product type
4  typename IntPair = (Int, Int) ;
5  # record type
6  typename Pet = (name:String, age:Int, species:String, Fur:Bool) ;
7  # function type
8  typename Comp(a, e::Row) = () ~e~> a ;

```

Listing 8. Type alias examples

But until now, there was no way to define such thing for effects. This is really usefull since when working with effects, the same set of effects show up everywhere.

Using the new keyword `effectname` we can now write an alias for an effect row :

```

1  effectname MyEffectRow(a, ... ,e::Eff, ...) = { Op1 : type, ... | e } ; # open row
2  effectname MyEffectRow(a, ... ) = { Op1 : type, ... } ; # closed row

```

To use it in a signature or in another type or effect alias just apply it with the right arguments as for `typename` things. In arrows, it can be used as row variables:

```

1  () -MyEffectRow(args)-> ()

```

However, due to lack of kind inference, row variables and aliases have to be used carefully so that Links does not think they are of kind type. We need to write them most of the time between braces `{ | ... }`. For instance, if there is `effectname E(a::Eff)= {X : ... | a }` and a row variable `e::Eff`, you will have to write `E({ |e})`, and similarly for another effect alias instead of the variable. This makes the usage of several nested aliases a bit messy, it would be nice if it could be avoided.

Recursive effect aliases are not available for now. The aliases are replaced by the row they correspond to, aliases are not kept are not printed out.

The implementation is mostly inspired by what is done for type aliases.

5.2 Fresh labels

5.2.1 The pollution problem. We first introduce a problem that arises when playing with numerous effects and that we will be calling *effect pollution*. Let us start with an example.

```

1  sig receives : ([Int]) -> ( Comp(a, {Fail:Zero, Receive:Int |e}) ) -> Comp(a , {Fail:
    Zero, Receive{_|e}})
2  fun receives (ss) (h) () {
3      handle (h()) (ss<-ss) {
4          case <Receive => r> -> switch (ss) {
5              case [] -> absurd(do Fail)
6              case h::t -> r (h,t)
7          }
8          case x -> x
9      }
10 }

```

Let receives be a (parameterised) handler for the operation `Receive` that provides an integer. This handler will give the integers from a list that it has been given. When the end of the list is reached, an error is raised.

```

1  fun not_too_big () {
2      var x = do Receive ;
3      if (x >= 42)
4          absurd(do Fail)
5      else x
6  }
7
8  typename Sized(a) = (val:Maybe(a), big:Bool) ;
9
10 sig solve_too_big : ( Comp(a, {Fail:Zero |e}) ) -> Comp(Sized(a), {Fail{_|e}})
11 fun solve_too_big (h) () {
12     handle (h()) {
13         case <Fail> -> (val=Nothing, big=true)
14         case x      -> (val=Just(x), big=false)
15     }
16 }

```

Let us imagine another piece of code that uses this effect `Receive` but does not want too big numbers. When something too big is given, a failure is raised and then catch by a handler that will keep track of it.

What happen if we want to have these work together ?

```

1  fun two () { not_too_big () + not_too_big () }
2
3  links> solve_too_big( maybe( receives ([21,21,21]) ( two ))) () ;
4  (big = false, val = Just(Just(42))) : Sized (Maybe (Int))
5
6  links> solve_too_big( maybe( receives ([42,42,42]) ( two ))) () ;
7  (big = false, val = Just(Nothing)) : Sized (Maybe (Int))

```

If we compose handlers the way above, we see that `666` is too big but the `big` value is set to `false`. The `Fail` raised by `not_too_big` has been catch by `maybe`. The shared `Fail` effects interfere and are mixed up. This is what we call *effect pollution*.

That kind of effect pollution could become really annoying when one works with a large amount of code and numerous effects and handlers and it reduces modularity.

5.2.2 State of the art. Until then, we could find two constructions that tackle this issue in the literature : *lexically scoped handlers* and *effect coercions*.

Effect coercions. Effect coercions, also known as *adaptators* ([CLMM20] section 3), are constructions that change how operations are interpreted and handled.

The simplest example is the *lift* or *mask* coercion that masks the operation from the nearest handler. A masked operation is then handled by the second nearest handler.

Coercions could also be exchanges, also called *swap*, or effect duplication in some contexts.

It turns out that, if they offer a solution to the pollution problem, they are not so practical. Moreover they seems a bit *ad hoc* and are not really satisfying.

Lexically scoped handlers. [SBMO22] introduces this kind of handlers where one can figure out statically which handler is invoked when. This is achieved by not having operations but computations that take an operation function as parameter.

```
1  lexhandle ( fun (op) { e } ) with { case <p => r> -> ... }
```

Remark 5.1. In this setting, the handler does not handle “by name”. That’s why it does only need the payload and not the label of the operation in `case <p => r>`.

Biernacki et al. [BPPS19] use another version of lexically scoped effects with names. Handlers and operations labels are coupled with names that binds them together. Handlers act as binders for theses names.

5.2.3 A fresh idea ? In [EdVP22], de Vilhena and Pottier develop the idea of lexically scoped handlers and propose, as we do, to use fresh effect label generation. Our work has been done in parallel but it turns out that they formalised a system close to what we use and describe there.

Our system has two sort of labels, noted l :

global labels labels as previously used, defined uniquely by their name

local labels labels that are restricted to a scope and are identified by their name together with an identifier

$$id \in \mathbb{N} \cup \{\perp\}$$

$$l ::= Op \mid 'Op^{id}$$

To distinguish local and global labels we use a backtick for local ones. The programmer does not need to find by themselves the identifier. They do not specify any (which is represented by the “free” identifier \perp) and fresh unique identifiers are created automatically.

We define a new construct **fresh** $\backslash A \{ e \}$ where e is any list of function and/or aliases definitions. This is a binding for local labels. When this is encountered a new unique identifier is generated for $\backslash A$ and each local occurrence of $\backslash A$ in the scope e is bound to this identifier.

Remark 5.2. When several **fresh** bindings with the same name are nested, a label is bound by the innermost **fresh**.

Going out of scope. A function, value or alias that has a local label in its type cannot escape the scope, but we do want to keep some of the declarations made in that scope. When leaving the scope, for each declaration, we try to *erase* the fresh label l from its type. If it is possible, ie if the label has a polymorphic or absent presence, the function/value/alias declared is added into the context and is available in the rest of the program, its type being cleaned from any local label. Otherwise, the function/value/alias is removed from the context.

We note $A \uparrow_l A'$ if one can erase label l from type A and this will give type A' , and $A \downarrow_l$ if it is not possible to erase label l from type A , ie if there exists no A' such that $A \uparrow_l A'$. Below are the most interesting rules that define the erasure of a row type. The rest of the rules can be found in appendix D.

$$\frac{R \uparrow_l R'}{l : -; R \uparrow_l R'} \quad \frac{R \uparrow_l R'}{l : \emptyset; R \uparrow_l R'} \quad \frac{T \uparrow_l T' \quad R \uparrow_l R' \quad l \neq l'}{l' : T; R \uparrow_l l' : T'; R'} \quad \frac{}{l : T; R \downarrow_l}$$

The two first rules state that absent or polymorphic instances of the label can be erased, the third one shows how the erasure propagates into the row and the last one forces the erasure of a present instance to fail.

Theorem 5.3. *For any label l and type T , exactly one of the following holds*

- ◊ *either there exists a unique T' such that $T \uparrow_l T'$. we write $e(T) = T'$*
- ◊ *or $T \downarrow_l$*

At the end of the scope, for each declaration, we type it and get its type T . If $e(T)$ exists, we replace its type by $e(T)$ in the context. Otherwise, we remove the object from the context. Thus, local labels cannot escape.

5.2.4 Implementation and usage in Links. The possibility to define local labels, ie labels that can only be used in a specific scope, has been added to Links.

The user can define one or more local labels that will be bound in the scope with

```

1  fresh  $\backslash A, \backslash B \{$ 
2      [ scope where labels  $\backslash A$  and  $\backslash B$  are bound ]
3  }
```


He can use the local labels so defined in the scope as any other effect label but he cannot use them outside that scope.

Solving pollution. This can be used to solve pollution issues by shadowing effects that would interfere with a local name.

```

1  fresh `Fail {
2
3      sig protect : (Comp(a, {`Fail:Zero, Fail:Zero | e})) -> Comp(a, {Fail{`_, `Fail:Zero |
        e})
4      fun protect (m) () {
5          handle (m()) {
6              case <Fail => k>-> k (do `Fail)
7          }
8      }
9
10     sig unprotect : (Comp(a, {`Fail:Zero, Fail:Zero | e})) -> Comp(a, {`Fail{`_, Fail:
        Zero|e})
11     fun unprotect (m) () {
12         handle (m()) {
13             case <`Fail => k> -> k (do Fail)
14         }
15     }
16
17     typename CNonLoc(a,e::Eff) = Comp(a, {Receive:Int, Fail:Zero, `Fail-    |e}) ;
18     typename CLoc(a,e::Eff)    = Comp(a, {Receive:Int, Fail:Zero, `Fail:Zero|e}) ;
19
20     sig protected_receives : ([Int]) -> (CNonLoc(a,{ |e})) -> Comp(Maybe(a), {Receive{`_,
        Fail:Zero, `Fail{`_} |e})
21     fun protected_receives (xs) (h) { unprotect ( maybe( receives ([xs]) ( protect ( h :
        CLoc(a,{ |e}) <- (CNonLoc(a,{ |e}))
22     ) ) ) ) }
23
24 }
25
26 links> protected_receives ;
27 protected_receives = fun : ([Int]) -> (Comp(a,{Receive:Int, Fail:Zero |e})) -> Comp(
        Maybe(a), {Receive{`_, Fail:Zero |e})
28
29 links> solve_too_big( protected_receives ([21,21,21]) ( two ))) () ;
30 (big = false, val = Just(Just(42))) : Sized (Maybe (Int))

```

```

31
32   links> solve_too_big( protected_receives ([42,42,42]) ( two ))) () ;
33   (big = true, val = Nothing) : Sized (Maybe (Int))

```

Listing 9. Using local effects to avoid pollution

We can declare a protected version of `receives` that guaranties that other `Fail` effects will not interfere. When a “too big” number is encountered, the `Fail` effect raised can go through `protected_receives`, reach `solve_too_big` and be correctly handled. The `big` field is set to `true`.

Remaining issues. There is one issue remaining. The effect aliases being inlined before type-checking, we can get a label going out of scope. This will be identified by the typechecker and will raise a `local label ... is not bound` error. It seems that the `effectname` has escaped the scope.

```

1   fresh `A {
2       effectname E = {`A: Int} ;
3   }
4   typename T = () -E-> () ;
5
6   links>
7   Type error: The local label `A<1> is not bound
8   In expression: typename T = () -E-> ().

```

This is not really satisfying but should be fixed by improving effect aliases and giving them the same behavior as type aliases. Furthermore, this could not cause code that should not type-check to type-check since it is catch.

Implementation. Briefly, here is how the code works. All labels in the code can be either global or local, have a name and the local ones have a unique id. Labels with a backtick are parsed into free local label and normal label are parsed into global ones. During desugaring, a `newpass` binds the local labels. During type-checking, when a `fresh ... { ... }` binding is encountered, we add the labels into the context and typecheck each declaration in the scope. Before that, the context should be clean from any label that will be shadowed by the fresh declaration (labels that have the same name). When leaving the scope, we clean the context from the local labels to avoid them to escape the scope. Moreover we check during typechecking if types do not contain unbound or shadowed labels.

Remark 5.4. For now if all labels are internally of the same type and then can be local or global, the parser only accepts local labels in effect related things. If needed, this could be a way of improvement.

5.3 Polymorphic operations

My last contribution was to implement the support for polymorphic operations in Links. Until then, only handlers could be polymorphic but not the operations themselves. However this could be useful in a lot of

situations. For instance, when writing exceptions, one would want to have a polymorphic `Fail` operation so that it could be used in any location. Without actual polymorphism this could be simulated with a `Zero` type and its destructor as we've done in previous examples.

This is easily added since polymorphism is already incorporated in Links. The only thing to add is to allow operations to be of the form $\forall \alpha :: K \dots$ and not only fat arrow types.

To make an operation polymorphic, the handler should be annotated with a polymorphic type. The type system cannot infer polymorphism and is not able to propagate the polymorphism information in the signature down to the effect cases. This is something that could be improved in the future.

```

1  sig catch : (() {Fail:forall a.a |e}~> b) {Fail{_|e}~> Maybe(b)
2  fun catch(m) {
3      handle(m()) {
4          case <Fail => k> : (forall a. () => a) -> Nothing
5          case x -> Just(x)
6      }
7  }
8
9  sig f : () {Fail:forall a.a}~> Int
10 fun f () {
11     if (do Fail)
12         42
13     else
14         do Fail
15 }
16
17 catch (f)
18 links> Nothing : Maybe (Int)

```

Listing 10. Polymorphism example

6 CONCLUSION

In this internship, I have discovered a new programming language, Links, and a new feature, effect handlers. I learnt how they work and I tried to make a contribution.

I have incorporated in Links useful and interesting features but they are far from perfect and each of them could and will be improved in future versions of Links. For instance, type and effect aliases could be generalised and the kinding system might need some improvements for this. One may also want to go towards first class labels. I provided a way of creating new instances of labels but this could be extended even if this raises a huge amount of new issues. The way we achieve polymorphic operations is a bit constraining because the annotations have to be at the level of cases. Using some sort of bidirectionality, one may hope

to avoid this and allow the programmer to only write polymorphic types in the signature of the handler for instance.

This internship allowed me to face many challenges related to code development such as working with a large codebase, understanding not-so-much-commented code, dealing with git. It is really satisfying to see that the code I wrote is now running and will be running in the official Links code.

In the end, I am really glad to have been able to have this experience and that my work could be, at least a little, useful to research.

Acknowledgements. I would like to thank the team that welcomed me in Edinburgh, that is of course Sam and Daniel my supervisors but also all the members of the PL group. I also would like to thank my fellow interns Nachi, Yi and Pierre and my flatmates Aleks, Iris and Misha that had to put up with me during these few months in Edinburgh.

REFERENCES

- [BPPS19] Dariusz Biernacki, Maciej Piróg, Piotr Polesiuk, and Filip Sieczkowski. Binders by day, labels by night: Effect instances via lexically scoped handlers. *Proc. ACM Program. Lang.*, 4(POPL), dec 2019.
- [CCFL19] Kwanghoon Choi, James Cheney, Simon Fowler, and Sam Lindley. A polymorphic RPC calculus. *CoRR*, abs/1910.10988, 2019.
- [CLMM20] Lukas Convent, Sam Lindley, Conor McBride, and Craig McLaughlin. Doo bee doo bee doo. *Journal of Functional Programming*, 30:e9, 2020.
- [CLWY07] Ezra Cooper, Sam Lindley, Philip Wadler, and Jeremy Yallop. Links: Web programming without tiers. In Frank S. de Boer, Marcello M. Bonsangue, Susanne Graf, and Willem-Paul de Roever, editors, *Formal Methods for Components and Objects*, pages 266–296, Berlin, Heidelberg, 2007. Springer Berlin Heidelberg.
- [EdVP22] Paulo Emílio de Vilhena and François Pottier. A type system for effect handlers and dynamic labels. 2022.
- [GTL89] Jean-Yves Girard, Paul Taylor, and Yves Lafont. *Proofs and Types*. Cambridge University Press, USA, 1989.
- [Hil22] Daniel Hillerström. *Foundations for programming and implementing effect handlers*. PhD thesis, 2022.
- [HL16] Daniel Hillerström and Sam Lindley. Liberating effects with rows and handlers. In *Proceedings of the 1st International Workshop on Type-Driven Development*, TyDe 2016, page 15–27, New York, NY, USA, 2016. Association for Computing Machinery.
- [HLA20] Daniel Hillerström, Sam Lindley, and Robert Atkey. Effect handlers via generalised continuations. *Journal of Functional Programming*, 30:e5, 2020.
- [Joua] Robin Jourde. Links fork. <https://github.com/Orbion-J/links>.
- [Joub] Robin Jourde. Links pull requests. <https://github.com/links-lang/links/issues?q=author%3AOrbion-J+>.
- [LC12] Sam Lindley and James Cheney. Row-based effect types for database integration. In *Proceedings of the 8th ACM SIGPLAN Workshop on Types in Language Design and Implementation*, TLDI ’12, page 91–102, New York, NY, USA, 2012. Association for Computing Machinery.
- [PP09] Gordon Plotkin and Matija Pretnar. Handlers of algebraic effects. In Giuseppe Castagna, editor, *Programming Languages and Systems*, pages 80–94, Berlin, Heidelberg, 2009. Springer Berlin Heidelberg.
- [Ré94] Didier Rémy. Type inference for records in a natural extension of ml. *Theoretical Aspects of Object-Oriented Programming: Types, Semantics, and Language Design*, 1994.
- [SBMO22] Philipp Schuster, Jonathan Immanuel Brachthäuser, Marius Müller, and Klaus Ostermann. A typed continuation-passing translation for lexical effect handlers. In *Proceedings of the 43rd ACM SIGPLAN International Conference on Programming*

Language Design and Implementation, PLDI 2022, page 566–579, New York, NY, USA, 2022. Association for Computing Machinery.

- [Weba] Web. Effect handlers for webassembly. <https://wasmfx.dev/>.
- [Webb] Web. Links repository. <https://github.com/links-lang/links>.
- [Webc] Web. Links website. <https://www.links-lang.org>.
- [Webd] Web. Pyro webpage. <https://pyro.ai/>.
- [Webe] Web. Pyro’s poutine library. <https://docs.pyro.ai/en/stable/poutine.html>.

APPENDIX

A ARTEFACT

The code that I have produced during this internship is available in my Links fork on GitHub [Joua]. The notable branches are the master, the `fresh-label` branch that contains the work described in section 5.2 and the `polymorphic-operations` branch that contains the work described in section 5.3.

The pull requests that I opened are available here [Joub].

The Links repository can be found at [Webb]. Set up and installation instructions are available on Links website [Webc] or you can built in from source using the Makefile provided in the repository. After having installed the dependencies with `$ opam install --deps-only links`, a simple `$ make` should be sufficient.

To replicate the examples in this paper, use preferably the branches of my fork. The changes will be step by step included in Links and in new releases.

B LINKS TECHNICALITIES

Here are some technical points on Links syntax and semantics that could be useful to the reader.

B.1 Comments

Everything following a `#` in a line is a comment.

B.2 Tuples

A tuple is a closed record with positive integer labels. For instance `(foo, bar)` is the same as `(1=foo, 2=bar)`.

Unit `()` is the empty record.

B.3 Operation type

An operation should always be of type `(a) => b` but one can write `Op:b` which will be desugared into `Op:() => b`.

A polymorphic operation type can also be sugared. One can write `Op:forall a. b` for `Op:forall a. ()=> b`. It is worth noting that this does *not* stand for `Op:()=> forall a. b`.

B.4 Empty type

The empty variant type `[]` is Zero. To destroy it, we use the following `absurd` function.

```

1  sig absurd : (Zero) -> a
2  fun absurd(x) {
3      switch ( x ) {}
4  }
```

B.5 Option type

Links has a builtin polymorphic option type `Maybe` defined as below. An object of type `Maybe(a)` is either `Nothing` or `Just(x)` where `x` is of type `a`

```
1  typename Maybe(a) = [| Nothing | Just : a |]
```

B.6 Function types

Function types are written `(a) {...} → b`. The domain type is always embraced with parentheses. The braces contains an effect row that tells among others which effects can be performed by the function.

One can write `(a) -e→ b` when `e` is a row variable or an effect type and `(a) → b` stands for `(a) -_→ b` : the function is then polymorphic in its effects.

`(a) {...} → b`, `(a) ~e→ b` or `(a) ~> b` is a syntactic sugar that adds the special effect `wild:()` to the row. This is used to denote “wild” code, ie code that cannot be transformed into a SQL query. Recursion, IO or handlers are typically wild. See [LC12] for justifications and details on the use of row typing for database integration.

For more details on row typing, see section 4.1.

B.7 Computation and handlers

Links has a builtin type for computations

```
1  typename Comp(a, e::Row) = () ~e~> a
```

For better handler composability, we write handlers that take a computation and output another computation. Then we can compose them easily and we just need to launch the computation at the end :

```
1  # computation
2  handler''( handler'( handler( f ) ) )
3  # value
4  handler''( handler'( handler( f ) ) ) ()
```

C TYPING RULES

We give extra typing rules for shallow and parameterised handlers.

$$\frac{\Delta; \Gamma \vdash m : B/R' \quad \Delta; \Gamma, x : B \vdash n : A/R \quad \begin{array}{c} R' = l_i : B_i \Rightarrow A_i; R_0 \quad R = l_i : \theta; R_0 \\ \Delta; \Gamma, p_i : B_i, r_i : A_i \rightarrow_{R'} A \vdash n_i : A/R \end{array}}{\Delta; \Gamma \vdash \text{shallowhandle}(m)\{\text{case } x \mapsto n \text{ case } \langle l_i(p_i) \Rightarrow r_i \rangle \mapsto n_i\} : A/R} \text{ShallowHandle}$$

$$\frac{\Delta; \Gamma \vdash m : B/R' \quad \Delta; \Gamma \vdash w : \gamma/R' \quad \begin{array}{c} R' = l_i : B_i \Rightarrow A_i; R \quad R = l_i : \theta; R_0 \\ \Delta; \Gamma, x : B, s : \gamma \vdash n : A/R \quad \Delta; \Gamma, s : \gamma, p_i : B_i, r_i : (A_i, \gamma) \rightarrow_{R'} A \vdash n_i : A/R \end{array}}{\Delta; \Gamma \vdash \text{handle}(m)(s \leftarrow w)\{\text{case } x \mapsto n \text{ case } \langle l_i(p_i) \Rightarrow r_i \rangle \mapsto n_i\} : A/R} \text{ParamHandle}$$

D ERASING PROCEDURE

We give the missing rules for the erasure procedure introduced in section 5.2.3.

$$\frac{x = \alpha, P, \rho, \bullet}{x \uparrow_l x} \quad \frac{A \uparrow_l A' \quad B \uparrow_l B' \quad R \uparrow_l R'}{A \rightarrow_R B \uparrow_l A' \rightarrow_{R'} B'}$$

$$\frac{A \uparrow_l A' \quad B \uparrow_l B'}{A \Rightarrow B \uparrow_l A' \Rightarrow B'} \quad \frac{A \uparrow_l A' \quad B \uparrow_l B'}{A B \uparrow_l A' B'} \quad \frac{T \uparrow_l T'}{\forall \alpha :: K.T \uparrow_l \forall \alpha :: K.T'}$$

$$\frac{A \downarrow_l}{A \rightarrow_R B \downarrow_l} \quad \frac{B \downarrow_l}{A \rightarrow_R B \downarrow_l} \quad \frac{R \downarrow_l}{A \rightarrow_R B \downarrow_l}$$

$$\frac{A \downarrow_l}{A \Rightarrow B \downarrow_l} \quad \frac{B \downarrow_l}{A \Rightarrow B \downarrow_l} \quad \frac{A \downarrow_l}{A B \downarrow_l} \quad \frac{B \downarrow_l}{A B \downarrow_l} \quad \frac{T \downarrow_l}{\forall \alpha :: K.T \downarrow_l}$$